

Lösung 5

Aufgabe 22

```
; BUBBLE SORT, STEIGEND
; 21 ZAHLEN
; AEUSSERE SCHLEIFE (I)
; INNERE SCHLEIFE (J)
; VERGLEICHEN
M1:   LDA      ZAHL_2      ; A := <ZAHL_2>      <-- MODIFIZIEREN !
      MAB      ; B := A                      (B := ZAHL_2)
M2:   LDA      ZAHL_1      ; A := <ZAHL_1>      <-- MODIFIZIEREN !
      SUB      ; A := A - B                    (A := ZAHL_1 - ZAHL_2)
      JPS      M5          ; IF <ZAHL_1> < <ZAHL_2> GOTO M5
; TAUSCHEN
      ADD      ; A := A + B                    (A := ZAHL_1)
M3:   MOV      ZAHL_2      ; <ZAHL_2> := A      <-- MODIFIZIEREN !
      LDA      NULL       ; A := 0
      ADD      ; A := A + B                    (A := ZAHL_2)
M4:   MOV      ZAHL_1      ; <ZAHL_1> := A      <-- MODIFIZIEREN !
; BEFEHLSMODIFIKATION INNERE SCHLEIFE (ZAHL_2)
M5:   LDB      EINS        ; B := 1
      LDA      M1          ; A := <M1>
      ADD      ; A := A + B
      MOV      M1          ; <M1> := A
      LDA      M3          ; A := <M3>
      ADD      ; A := A + B
      MOV      M3          ; <M3> := A
; J DEKREMENTIEREN
      LDA      J           ; A := <J>
      SUB      ; A := A - B
      MOV      J           ; <J> := A          (J := J - 1)
      JPS      M6          ; IF A < 0 GOTO M6    (IF J < 0 GOTO M6)
      JMP      M1          ; GOTO M1
; BEFEHLSMODIFIKATION AEUSSERE SCHLEIFE (ZAHL_1, ZAHL_2)
M6:   LDA      M2          ; A := <M2>
      ADD      ; A := A + B
      MOV      M2          ; <M2> := A
      ADD      ; A := A + B
      MOV      M1          ; <M1> := A
      LDA      M4          ; A := <M4>
      ADD      ; A := A + B
      MOV      M4          ; <M4> := A
      ADD      ; A := A + B
      MOV      M3          ; <M3> := A
; I DEKREMENTIEREN, J = I SETZEN
      LDA      I           ; A := <I>
      SUB      ; A := A - B
      MOV      I           ; <I> := A          (I := I - 1)
      JPS      M7          ; IF A < 0 GOTO M7    (IF I < 0 GOTO M7)
      MOV      J           ; <J> := A          (J := I)
      JMP      M1          ; GOTO M1
;
M7:   HLT              ;
;
I:    DEF      19          ; I := ANZ - 2
J:    DEF      19          ; J := ANZ - 2
NULL: DEF      0          ; KONSTANTE 0
EINS: DEF      1          ; KONSTANTE 1
;
ZAHL_1: DEF     10         ; 21 ZAHLEN
ZAHL_2: DEF     9
      DEF     8
      DEF     7
      DEF     6
      DEF     5
```

```

DEF      4
DEF      3
DEF      2
DEF      1
DEF      0
DEF     -1
DEF     -2
DEF     -3
DEF     -4
DEF     -5
DEF     -6
DEF     -7
DEF     -8
DEF     -9
DEF    -10

```

Aufgabe 23

```

; BUBBLE SORT, FALLEND
; 21 ZAHLEN
; AEUSSERE SCHLEIFE (I)
; INNERE SCHLEIFE (J)
; VERGLEICHEN
M1:   LDA      ZAHL_1      ; A := <ZAHL_1>      <-- MODIFIZIEREN !
      MAB                      ; B := A          (B := ZAHL_1)
M2:   LDA      ZAHL_2      ; A := <ZAHL_2>      <-- MODIFIZIEREN !
      SUB                      ; A := A - B        (A := ZAHL_2 - ZAHL_1)
      JPS      M5          ; IF <ZAHL_2> < <ZAHL_1> GOTO M5
; TAUSCHEN
      ADD                      ; A := A + B        (A := ZAHL_2)
M3:   MOV      ZAHL_1      ; <ZAHL_1> := A        <-- MODIFIZIEREN !
      LDA      NULL        ; A := 0
      ADD                      ; A := A + B        (A := ZAHL_1)
M4:   MOV      ZAHL_2      ; <ZAHL_2> := A        <-- MODIFIZIEREN !
; BEFEHLSMODIFIKATION INNERE SCHLEIFE (ZAHL_2)
M5:   LDB      EINS        ; B := 1
      LDA      M2          ; A := <M2>
      ADD                      ; A := A + B
      MOV      M2          ; <M2> := A
      LDA      M4          ; A := <M4>
      ADD                      ; A := A + B
      MOV      M4          ; <M4> := A
; J DEKREMENTIEREN
      LDA      J           ; A := <J>
      SUB                      ; A := A - B
      MOV      J           ; <J> := A            (J := J - 1)
      JPS      M6          ; IF A < 0 GOTO M6      (IF J < 0 GOTO M6)
      JMP      M1          ; GOTO M1
; BEFEHLSMODIFIKATION AEUSSERE SCHLEIFE (ZAHL_1, ZAHL_2)
M6:   LDA      M1          ; A := <M1>
      ADD                      ; A := A + B
      MOV      M1          ; <M1> := A
      ADD                      ; A := A + B
      MOV      M2          ; <M2> := A
      LDA      M3          ; A := <M3>
      ADD                      ; A := A + B
      MOV      M3          ; <M3> := A
      ADD                      ; A := A + B
      MOV      M4          ; <M4> := A
; I DEKREMENTIEREN, J = I SETZEN
      LDA      I           ; A := <I>
      SUB                      ; A := A - B
      MOV      I           ; <I> := A            (I := I - 1)
      JPS      M7          ; IF A < 0 GOTO M7      (IF I < 0 GOTO M7)
      MOV      J           ; <J> := A            (J := I)
      JMP      M1          ; GOTO M1

```

```

;
M7:    HLT                ;
;
I:      DEF      19      ; I := ANZ - 2
J:      DEF      19      ; J := ANZ - 2
NULL:   DEF      0       ; KONSTANTE 0
EINS:   DEF      1       ; KONSTANTE 1
;
ZAHL_1: DEF      -10     ; 21 ZAHLEN
ZAHL_2: DEF      -9
        DEF      -8
        DEF      -7
        DEF      -6
        DEF      -5
        DEF      -4
        DEF      -3
        DEF      -2
        DEF      -1
        DEF      0
        DEF      1
        DEF      2
        DEF      3
        DEF      4
        DEF      5
        DEF      6
        DEF      7
        DEF      8
        DEF      9
        DEF      10

```

Aufgabe 24

```

; MISCHEN ZWEIER STEIGEND SORTIERTER FOLGEN
M1:    LDA      I        ; A := <I>, FLAG S SETZEN
        JPS     M2        ; IF <I> < 0 GOTO M2
        JMP     M3        ; GOTO M3
;
M2:    LDA      J        ; A := <J>, FLAG S SETZEN
        JPS     FERTIG    ; IF <J> < 0 GOTO FERTIG
        JMP     M8        ; GOTO M8
;
M3:    LDA      J        ; A := <J>, FLAG S SETZEN
        JPS     M6        ; IF <J> < 0 GOTO M6
;
M4:    LDA      ZAHL_2    ; A := <ZAHL_2>      <-- MODIFIZIEREN !
M5:    LDB      ZAHL_1    ; B := <ZAHL_1>      <-- MODIFIZIEREN !
        SUB     M8        ; A := <ZAHL_2> - <ZAHL_1>
        JPS     M8        ; IF <ZAHL_2> < <ZAHL_1> GOTO M8
;
M6:    LDA      ZAHL_1    ; A := <ZAHL_1>      <-- MODIFIZIEREN !
M7:    MOV      ZAHL_3    ; <ZAHL_3> := A      <-- MODIFIZIEREN !
        LDB     EINS      ; B := 1
        LDA     I         ; A := <I>
        SUB     I         ; A := A - B
        MOV     I         ; <I> := <I> - 1
        LDA     M5        ; A := <M5>
        ADD     M5        ; A := A + B
        MOV     M5        ; <M5> := <M5> + 1
        LDA     M6        ; A := <M6>
        ADD     M6        ; A := A + B
        MOV     M6        ; <M6> := <M6> + 1
        JMP     M10       ; GOTO M10
;
M8:    LDA      ZAHL_2    ; A := <ZAHL_2>
M9:    MOV      ZAHL_3    ; <ZAHL_3> := A
        LDB     EINS      ; B := 1

```

```

LDA      J      ; A := <J>
SUB      ; A := A - B
MOV      J      ; <J> := <J> - 1
LDA      M4     ; A := <M4>
ADD      ; A := A + B
MOV      M4     ; <M4> := <M4> + 1
LDA      M8     ; A := <M8>
ADD      ; A := A + B
MOV      M8     ; <M8> := <M8> + 1
;
M10:     LDA      M7     ; A := <M7>
ADD      ; A := A + B
MOV      M7     ; <M7> := <M7> + 1
MOV      M9     ; <M9> := <M9> + 1
JMP      M1     ; GOTO M1
;
FERTIG:  HLT      ; HALT
;
EINS:    DEF      1      ;
I:        DEF      3      ; M - 1
J:        DEF      4      ; N - 1
ZAHL_1:   DEF      1      ; QUELLFOLGE 1, LAENGE M = 4
DEF      5      ;
DEF      12     ;
DEF      27     ;
ZAHL_2:   DEF      2      ; QUELLFOLGE 2; LAENGE N = 5
DEF      7      ;
DEF      16     ;
DEF      28     ;
DEF      30     ;
ZAHL_3:   DEF      0      ; ZIELFOLGE; LAENGE 0 = M + N = 9
DEF      0      ;
DEF      0      ;
DEF      0      ;
DEF      0      ;
DEF      0      ;
DEF      0      ;
DEF      0      ;
DEF      0      ;

```